PME 3380 – Modelagem de Sistemas Dinâmicos

LISTA G

Tiago Vieira de Campos Krause 9836238

1. Obtenha o modelo de ½ carro:



$M = 200$ kg;
$J = 512$ kgm^2;
$l_A = 0{,}8$ m;
$l_B = 0{,}8$ m;
$k_A = 10.000$ N/m;
$k_B = 10.000$ N/m;
$b_A = 200$ Ns/m;
$b_B = 200$ Ns/m;
$v_H = 10$ m/s;

Modelo da dinâmica vertical:

A dinâmica referente ao movimento horizontal do centro de massa é desprezada, ou seja, a velocidade horizontal de G (v_H) é constante, logo o modelo deve ter 4 variáveis de estado:

- velocidade vertical v_G do centro de massa G.
- velocidade angular ω de AB em torno de G.
- elongação x_A da mola de rigidez k_A.
- elongação x_B da mola de rigidez k_B.

Entradas: velocidades verticais (v_C e v_D) dos pontos C e D.

Saídas: velocidade vertical v_G do centro de massa G e velocidade angular ω de AB em torno de G.

Hipóteses simplificadoras:

- Movimento apenas no plano da página.
- AC e BD permanecem sempre na vertical.
- Considere molas e amortecedores lineares.
- O deslocamento angular do segmento AB é pequeno (tal que $sen\alpha \cong tan\alpha \cong \alpha$ e $cos\alpha \cong 1$).

Para o sistema apresentado acima obtém-se as seguintes equações:

Força no ponto A:

$$F_A = -k_A x_A + b_A(v_C - v_A)$$

Força no ponto B:

$$F_B = -k_B x_B + b_B(v_D - v_B)$$

Equações da barra:

$$\ddot{x}_G = \frac{F_A + F_B}{M}$$

$$\ddot{\theta}_G = \frac{-F_A l_A + F_B l_B}{J}$$

Das condições da barra:

$$x_A = x_G - l_A \theta_G$$

$$x_B = x_G + l_B \theta_G$$

Logo:

$$v_A = v_G - l_A w$$

$$v_B = v_G + l_B w$$

Deste modo, o sistema pode ser escrito na forma:

$$\dot{x} = Ax + Bu$$

$$y = Cx + Du$$

Para os vetores x, y e u definidos no enunciado e pelas equações acima pode-se obter os valores das matrizes, sendo:

$$A = \begin{bmatrix} 0 & 0 & 1 & -l_A \\ 0 & 0 & 1 & l_B \\ -\frac{k_A}{M} & -\frac{k_B}{M} & -\frac{C_A + C_B}{M} & \frac{C_A l_A - C_B l_B}{M} \\ \frac{k_A l_A}{J} & -\frac{k_B l_B}{J} & \frac{C_A l_A - C_B l_B}{J} & \frac{C_A l_A^2 - C_B l_B^2}{J} \end{bmatrix}$$

$$B = \begin{bmatrix} 0 & 0 \\ 0 & 0 \\ \frac{c_A}{M} & \frac{c_B}{M} \\ -\frac{c_A l_A}{J} & \frac{c_B l_B}{J} \end{bmatrix}$$

$$C = \begin{bmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$D = \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

2. Simulação do modelo de ½ carro

Simule o sistema de suspensão para entrada do tipo degrau. Considere condições iniciais nulas e tempo de simulação de 4 segundos.

$$\mathbf{u} = \begin{bmatrix} v_C \\ v_D \end{bmatrix} \qquad v_C = \begin{cases} 0 & \text{se} \quad t < 0 \\ 1 & \text{se} \quad t \geq 0 \end{cases} \qquad v_D = \begin{cases} 0 & \text{se} \quad t < t_d \\ 1 & \text{se} \quad t \geq t_d \end{cases}$$

Explique o tipo de obstáculo físico que é representado pela entrada degrau, e explique por que a entrada v_D ocorre t_d segundos após a entrada v_C (deve-se calcular t_d antes de se fazer a simulação).

Mostre os gráficos das saídas pelo tempo.

Simule o sistema de suspensão para entrada do tipo seno. Considere condições iniciais nulas. Simule por tempo suficiente para mostrar cerca de 20 períodos.

Entradas (observe que são duas simulações diferentes): $v_C = v_D = \text{sen}(9{,}8995t)$
$v_C = -v_D = \text{sen}(4{,}9875t)$

Repita as simulações para valores maiores e menores de freqüência. Compare os resultados.
Mostre os gráficos das saídas pelo tempo.
Calcule os coeficientes de amortecimento, as freqüências naturais, as freqüências naturais amortecidas e as freqüências de ressonância.

Elaborou-se o seguinte código para a simulação do modelo:

```
clear
clc

// Definindo os parâmetros

M=200;
J=512;
la=0.8;
lb=0.8;
ka=10000;
kb=10000;
ca=200;
cb=200;
vh=10;

// Definindo o sistema linear

A=[0,0,1,-la;
0,0,1,lb;
-ka/M,-kb/M,-(ca+cb)/M,(ca*la-cb*lb)/M;
ka*la/J,-kb*lb/J,(ca*la-cb*lb)/J,-(ca*la^2+cb*lb^2)/J];
B=[0,0;0,0;ca/M,cb/M;-ca*la/J,cb*lb/J];
C=[0,0,1,0;0,0,0,1];
D=[0,0;0,0];

amort= syslin('c',A,B,C,D);

// Definindo condições iniciais

x0 = [0;0;0;0];

// Definindo tempo de simulação

ttotal = 4;
t= 0:ttotal/500:ttotal;
```

```
// Definindo vetor de entradas
/*
u0 = [0;0];
vc = ones(t);
vd = ones(t);
td = (la+lb)/vh;
for i = 1:td*500/ttotal
    vd(i) = u0(2);
end
*/
vc = sin(9.8995*t)
vd = sin(9.8995*t)
/*
vc = sin(4.9875*t)
vd = -sin(4.9875*t)
*/
u = [vc;vd];

// Simulando o sistema usando o comando csim

[y,x]=csim(u,t,amort,x0)

// Plottando a resposta do sistema

scf(1)
plot2d(t,y(1,:),1)
xtitle("Resposta da velocidade vertical Vg do centro de massa", "Tempo [s]", "Vg [m/s]")
scf(2)
plot2d(t,y(2,:),2)
xtitle("Resposta da velocidade angular w de AB em torno do centro de massa", "Tempo [s]"
, "w [rad/s]")
```

A entrada degrau definida no início do enunciado apresenta uma velocidade constante igual a 1 a partir do instante t=0, isso representa uma subida de inclinação constante.

A entrada v_D ocorre t_D segundos após a entrada v_C pois ela representa os obtáculos na roda traseira do carro, isso significa que na trajetória a roda dianteira (C) passa primeiro pelo obstáculo e a traseira em seguida, sendo que t_D depende da velocidade horizontal do carro e da distância entre as rodas.

Para as condições iniciais nulas e os parâmetros a seguir definidos no enunciado, os gráficos obtidos foram os seguintes:

Simule o sistema de suspensão para entrada do tipo degrau. Considere condições iniciais nulas e tempo de simulação de 4 segundos.

$$\mathbf{u} = \begin{bmatrix} v_C \\ v_D \end{bmatrix} \qquad v_C = \begin{cases} 0 & \text{se} \quad t < 0 \\ 1 & \text{se} \quad t \geq 0 \end{cases}$$

$$v_D = \begin{cases} 0 & \text{se} \quad t < t_d \\ 1 & \text{se} \quad t \geq t_d \end{cases}$$





Para as entradas $v_C = v_D = \sin(9.8995t)$ e $v_C = -v_D = \sin(4.9875t)$ o sistema diverge e o scilab apresenta o seguinte erro:

```
 lsode--  at t (=r1), mxstep (=i1) steps
necessary before reaching tout
      where i1 is :        500
      where r1 is :   0.2842195575306D-03
Excessive work done on this call (perhaps wrong jacobian type).
```

Para uma entrada senoidal sendo $v_C = \sin(2t)$ e $v_D = \sin(t)$ obteve-se a seguinte resposta:





3. Análise de resposta em freqüência

Obtenha os diagramas de Bode do sistema de suspensão e <u>interprete</u> os resultados.

Para a obtenção dos diagramas de bode do sistema utilizou-se o seguinte código:

```
FT=ss2tf(amort)
k=0
for i=1:2
    for j=1:2
        k=k+1
        scf(2+k)
        bode(FT(i,j))
    end
end
```

Com a função *ss2tf* foram obtidas 4 funções de transferência, sendo a 3ª delas com o termo polinomial s negativo no denominador, por isso os diagramas resultantes foram iguais para as outras e diferente para a 3ª, sendo eles os seguintes:

Para a 3ª função de transferência:



Para as demais funções de transferência:



4. Simulação de sistema não linear

Exemplo

Considere o seguinte modelo de ¼ de carro, sem a massa não suspensa (1 grau de liberdade):



Este modelo em particular é a parâmetros concentrados, e será um sistema de equações diferenciais ordinárias.

Realizando a simulação desse exemplo usando o código fornecido com algumas adaptações, obteve-se a seguinte resposta do sistema:




```
// Definicao da funcao que implementa as equacoes diferenciais do sistema
function [xdot]=sistema(t,x,entrada)
if (x(1)-x(2))<lc then xdot=[x(3);entrada(t);(-kB*(x(1)-x(2)-l)-b*(x(3)-entrada(t))-m*g)/m];
elseif (x(1)-x(2))>l then xdot=[x(3);entrada(t);-g];
else xdot=[x(3);entrada(t);(-kM*(x(1)-x(2)-l)-b*(x(3)-entrada(t))-m*g)/m];
end
endfunction
```

Código que define o sistema

```
// Definicao da funcao que implementa a entrada vG:
function[ut]=entrada(t)
if t<ti then
ut=0;
elseif t<(ti+lB/vc) then
ut=(hB*2*%pi*vc/(2*lB))*sin((vc*2*%pi/lB)*(t-ti));
else
ut=0
end
endfunction
```

Código que define a entrada

```
// Definicao do arquivo que implementa a simulacao:
clear all
// Carregar a funcao que implementa o modelo matematico do sistema
exec("D:\Poli\8 semestre\Modelagem\Lista 7\sistema.sci");
// Carregar a função que implementa a entrada
exec("D:\Poli\8 semestre\Modelagem\Lista 7\entrada.sci");
// Definir os valores dos parametros
m=250; // massa [kg]
b=1885; // constante de amortecimento [Ns/m]
g=9.8; // aceleracao da gravidade [m/s2]
kM=14213; // rigidez da mola [N/m]
kB=142130; // rigidez do batente [N/m]
l=0.4; // comprimento natural da mola [m]
lc=0.1; // comprimento da mola totalmente comprimida [m]
hB=0.2; // altura da lombada [m]
lB=2; // comprimento da lombada [m]
ti=0.1; // tempo percorrido ate atingir a lombada [s]
vch=35; // velocidade do carro [km/h]
vc=vch/3.6; // velocidade do carro [m/s]
x0=[l-m*g/kM;0;0]; // condições iniciais
// O valor l-m*g/kM reflete a posição de equilíbrio da suspensão quando apenas o peso esta atuando.
t0=0; // instante inicial
t=0:0.0001:2.8; // vetor de tempo
x=ode(x0,t0,t,list(sistema,entrada));
// Plotando a diferenca entre a coordenada da massa e a coordenada
// do solo menos o comprimento natural da mola (deflexao):
plot2d(t,x(1,:)-x(2,:)-l)
// Se este valor é negativo, a mola esta comprimida.
// Se este valor é positivo, o carro "descolou" do solo.
// Se este valor diminui ate lc-l metros (neste caso -0.3 m), o batente é atingido.
// Plotando xG:
plot2d(t,x(2,:),2)
// Usando a variavel do tipo 'lista':
T=list("Resposta transitoria da suspensao","Tempo t [s]","Solução [m]","Deflexao","xG(perfil da via)");
// Colocando uma legenda na parte superior esquerda da figura (parametro 4):
legends([T(4),T(5)],[1,2],4);
// Colocando um titulo na figura e nomeando os eixos:
xtitle(T(1),T(2),T(3));
```

Código usado para realizar a simulação

Exercício

Modele um sistema não linear de suspensão veicular do tipo ¼ de carro, incluindo a massa não suspensa (2 graus de liberdade), com três entradas, a velocidade v_G imposta pelo movimento do veículo, uma força de perturbação F e uma força de controle u. Implemente a simulação do sistema não linear (considerando as não linearidades do exemplo da suspensão de ¼ de carro sem massa suspensa, e adicionando a saturação da entrada u, etc.).



Descrevendo o sistema:

O sistema possui dois graus de liberdade, um para cada bloco. Já os blocos possuem comportamentos não lineares devido às condições das molas as quais estão conectados.

A mola que liga bloco de massa *m* (massa não suspensa) ao chão tem três casos possíveis, assim como no exemplo indicado acima. Já a mola que liga os blocos possui dois casos possíveis (um a menos que a outra mola pois ela está conectada nas duas extremidades). A seguir estão definidas as variáveis e parâmetros do sistema e os casos descritos.

Variáveis do sistema:

$v(t)$: Velocidade vertical da massa não suspensa

$x(t)$: Coordenada vertical da massa não suspensa

$v_s(t)$: Velocidade vertical da massa suspensa

$x_s(t)$: Coordenada vertical da massa suspensa

$v_G(t)$: Velocidade vertical do ponto de veículo em contato com o solo

$x_G(t)$: Coordenada vertical do solo

$F(t)$: Força vertical de perturbação no sistema aplicada na massa suspensa

$u(t)$: Força de controle sujeita a uma saturação

Parâmetros do sistema:

m: Massa do bloco de massa não suspensa

k: Rigidez da mola que conecta o bloco de massa não suspensa ao solo

k_B: Rigidez do batente da mola que conecta o bloco de massa não suspensa ao solo

l: Comprimento natural da mola que conecta o bloco de massa não suspensa ao solo

l_C: Comprimento da mola que conecta o bloco de massa não suspensa ao solo totalmente comprimida

m_s: Massa do bloco de massa suspensa

k_s: Rigidez da mola que conecta os blocos

k_{B_s}: Rigidez do batente da mola que conecta os blocos

l_s: Comprimento natural da mola que conecta os blocos

l_{C_s}: Comprimento da mola que conecta os blocos totalmente comprimida

b: Constante de amortecimento do amortecidor que conecta os blocos

Situações para a mola que conecta o bloco de mola não suspensa ao solo:

a) Não há contato com o solo

Condição: $x - x_G > l$

Equação:

$$m\dot{v} = k'(x_s - x - l_s) + b(v_s - v) - mg - u$$

Sendo que k' depende da situação da mola que conecta os blocos

b) Existe contato com o solo mas ainda não se atingiu o batente

Condição: $l_C < x - x_G < l$

Equação:

$$m\dot{v} = k'(x_s - x - l_s) + b(v_s - v) - k(x - x_G - l) - mg - u$$

c) Existe contato com o solo e o batente foi atingido

Condição: $x - x_G < l_C$

Equação:

$$m\dot{v} = k'(x_s - x - l_s) + b(v_s - v) - k_B(x - x_G - l) - mg - u$$

Situações para a mola que conecta os blocos:

1) Mola comprimida ao máximo

Condição: $x_s - x < l_{C_s}$

Equação:

$$m\dot{v}_s = -k_{B_s}(x_s - x - l_s) - b(v_s - v) - mg + u - F$$

2) Mola não comprimida ao máximo

Condição: $x_s - x > l_{C_s}$

Equação:

$$m\dot{v}_s = -k_s(x_s - x - l_s) - b(v_s - v) - mg + u - F$$

Sendo que para a situação 1 $k' = k_{B_s}$ e para a situação 2 $k' = k_s$.

Definição das entradas:

Existem 3 tipos de entradas considerados no problema, a velocidade v_G e as forças $F(t)$ e $u(t)$. A seguir serão definidos os seus tipos de entrada e respectivos valores.

Velocidade v_G:

Será usada a mesma entrada que foi usada no exemplo anterior simulando uma lombada de comprimento l_B, altura h_B, e de formato senoidal.



Sendo ela descrita pelo mesmo código que definia a entrada no exemplo anterior.

Força $F(t)$:

Serão utilizadas entradas senoidais e do tipo degrau para simular o movimento das massas internas do veículo após passar a lombada. Sendo a entrada senoidal uma representação da massa cuja normal diminui quando a trajetória do veículo é de descida na vertical mas ainda é expressiva, já a entrada de degrau é uma representação da massa que possui normal nula (deixa de estar apoiada no veículo) durante a variação no sentido da trajetória vertical e em seguida volta a ser apoiada.

Para que sejam devidamente representadas, elas devem ser aplicadas aproximadamente no instante que o movimento do veículo é para baixo. Considerou-se esse instante como sendo o momento em que o veiculo passa por s_f da lombada.

Será considerado uma força de intensidade 600N, representando uma massa de 60kg no interior do veículo.

Força $u(t)$:

Será usada uma força senoidal para representar a saturação de entrada, tendo ela amplitude frequência unitária. Ela será aplicada a partir do momento em que a força $F(t)$ for aplicada.

Códigos:

Sendo o vetor de estados definido por:

$$x = [x \quad x_s \quad v \quad v_s \quad x_G]^T$$

Utilizou-se os seguintes códigos para a realização da simulação do sistema.

```
// Definicao da funcao que implementa as equacoes diferenciais do sistema
function [xdot]=sistema(t,x,entrada)
    [F,u,vG] = entrada(t);
    if lcs<x(2)-x(1)
        if l<x(1)-x(5)
            xdot = [x(3);x(4);(ks*(x(2)-x(1)-ls)+b*(x(4)-x(3))-u)/m-g;(-ks*(x(2)-x(1)-ls)-b*(x(4)-x(3))+u-F)/m-g;vG];
        elseif x(1)-x(5)<lc
            xdot = [x(3);x(4);(ks*(x(2)-x(1)-ls)+b*(x(4)-x(3))-kB*(x(1)-x(5)-l)-u)/m-g;(-ks*(x(2)-x(1)-ls)-b*(x(4)-x(3))+u-F)/m-g;vG];
        else
            xdot = [x(3);x(4);(ks*(x(2)-x(1)-ls)+b*(x(4)-x(3))-k*(x(1)-x(5)-l)-u)/m-g;(-ks*(x(2)-x(1)-ls)-b*(x(4)-x(3))+u-F)/m-g;vG];
        end
    else
        if l<x(1)-x(5)
            xdot = [x(3);x(4);(kBs*(x(2)-x(1)-ls)+b*(x(4)-x(3))-u)/m-g;(-kBs*(x(2)-x(1)-ls)-b*(x(4)-x(3))+u-F)/m-g;vG];
        elseif x(1)-x(5)<lc
            xdot = [x(3);x(4);(kBs*(x(2)-x(1)-ls)+b*(x(4)-x(3))-kB*(x(1)-x(5)-l)-u)/m-g;(-kBs*(x(2)-x(1)-ls)-b*(x(4)-x(3))+u-F)/m-g;vG];
        else
            xdot = [x(3);x(4);(kBs*(x(2)-x(1)-ls)+b*(x(4)-x(3))-k*(x(1)-x(5)-l)-u)/m-g;(-kBs*(x(2)-x(1)-ls)-b*(x(4)-x(3))+u-F)/m-g;vG];
        end
    end
endfunction
```

Código da função que define o sistema

```
// Definicao da funcao que implementa as entradas
function[F,u,vG]=entrada(t)
    if t<ti
        vG = 0;
        u = 0;
        F = 0;
    elseif t<tB
        vG =(hB*2*%pi*vc/(2*lB))*sin((vc*2*%pi/lB)*(t-ti));
        u = 0;
        F = 0;
    else
        vG = 0;
        u = 1000*sin(t-tB);
        // F degrau
        F = 600;
        // F senoidal
        //F = 600*sin(t-tB);
    end
endfunction
```

Código da função que define as entradas

```
// Definicao do arquivo que implementa a simulacao:
clear
clc
// Carregar a funcao que implementa o modelo matematico do sistema
exec("D:\Poli\8 semestre\Modelagem\Lista 7\sistema2.sci");
// Carregar a função que implementa a entrada
exec("D:\Poli\8 semestre\Modelagem\Lista 7\entrada2.sci");
// Definir os valores dos parametros
m = 250; // massa [kg]
b = 1885; // constante de amortecimento [Ns/m]
g = 9.8; // aceleracao da gravidade [m/s2]
k = 2*14213; // rigidez da mola da massa nao suspensa [N/m]
kB = 2*142130; // rigidez do batente da massa nao suspensa [N/m]
l = 0.4; // comprimento natural da mola da massa nao suspensa [m]
lc = 0.1; // comprimento da mola da massa nao suspensa totalmente comprimida [m]
ks = 14213; // rigidez da mola entre as massas [N/m]
kBs = 142130; // rigidez do batente da mola entre as massas [N/m]
ls = 0.4; // comprimento natural da mola entre as massas [m]
lcs = 0.1; // comprimento da mola entre as massas totalmente comprimida [m]
hB = 0.2; // altura da lombada [m]
lB = 2; // comprimento da lombada [m]
ti = 0.1; // tempo percorrido até atingir a lombada [s]
vch = 35; // velocidade do carro [km/h]
vc = vch/3.6; // velocidade do carro [m/s]
tB = ti+lB/vc; // tempo parcorrido até passar a lombada [s]
x0 = [l-2*m*g/k;l-2*m*g/k+ls-m*g/ks;0;0;0]; // condicoes iniciais
// O valor l-m*g/kM reflete a posicao de equilibrio da suspensao quando apenas o peso esta atuando.
t0 = 0; // instante inicial
t = t0:0.0001:2.8; // vetor de tempo
x = ode(x0,t0,t,list(sistema,entrada));
// Plotando a diferenca entre a coordenada da massa suspensa e a massa nao suspensa menos o comprimento natural da mola (deflexao):
figure
plot2d(t,x(1,:)-x(5,:)-l,1);
plot2d(t,x(2,:)-x(1,:)-ls,2);
plot2d(t,x(5,:),3);
T = list("Resposta Transitoria do Sistema","Tempo t [s]","Solucao [m]","Massa nao suspensa","Massa suspensa","Solo");
legends([T(4),T(5),T(6)],[1,2,3],1);
xtitle(T(1),T(2),T(3));
// Plotando a trajetoria das massas suspensa e nao suspensa
figure
plot2d(t,x(1,:),1);
plot2d(t,x(2,:),2);
T = list("Trajetoria das Massas","Tempo t [s]","Trajeto [m]","Massa nao suspensa","Massa suspensa");
legends([T(4),T(5)],[1,2],1);
xtitle(T(1),T(2),T(3));
```

Código para a realização da simulação

Resultados:

Realizando a simulação com as condições iniciais e parâmetros definidas no código acima, baseadas principalmente no exemplo feito anteriormente obteve-se os seguintes resultados.

Para entrada em degrau:





O gráfico de baixo representa a deflexão das molas, sendo a curva preta a mola que conecta a massa não suspensa com o solo e a curva azul a mola que conecta os blocos.

Para entrada senoidal: